F08KUF (CUNMBR/ZUNMBR) – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

F08KUF (CUNMBR/ZUNMBR) multiplies an arbitrary complex matrix C by one of the complex unitary matrices Q or P which were determined by F08KSF (CGEBRD/ZGEBRD) when reducing a complex matrix to bidiagonal form.

2 Specification

```
      SUBROUTINE F08KUF(VECT, SIDE, TRANS, M, N, K, A, LDA, TAU, C, LDC,
     1                  WORK, LWORK, INFO)
      ENTRY      cunmbr(VECT, SIDE, TRANS, M, N, K, A, LDA, TAU, C, LDC,
     1                  WORK, LWORK, INFO)
      INTEGER           M, N, K, LDA, LDC, LWORK, INFO
      complex           A(LDA,*), TAU(*), C(LDC,*), WORK(LWORK)
      CHARACTER*1       VECT, SIDE, TRANS
```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine is intended to be used after a call to F08KSF (CGEBRD/ZGEBRD), which reduces a complex rectangular matrix A to real bidiagonal form B by a unitary transformation: $A = QBP^H$. F08KSF represents the matrices Q and P^H as products of elementary reflectors.

This routine may be used to form one of the matrix products QC, Q^HC, CQ, CQ^H, PC, P^HC, CP or CP^H, overwriting the result on C (which may be any complex rectangular matrix).

4 References

[1] Golub G H and van Loan C F (1996) *Matrix Computations* Johns Hopkins University Press (3rd Edition), Baltimore

5 Parameters

In the description below, r denotes the order Q or P^H : $r = M$ if SIDE = 'L' and $r = N$ if SIDE = 'R'.

1: VECT — CHARACTER*1 *Input*

On entry: indicates whether Q or Q^H or P or P^H is to be applied to C as follows:

if VECT = 'Q', then Q or Q^H is applied to C;
if VECT = 'P', then P or P^H is applied to C.

Constraint: VECT = 'Q' or 'P'.

2: SIDE — CHARACTER*1 *Input*

On entry: indicates how Q or Q^H or P or P^H is to be applied to C as follows:

if SIDE = 'L', then Q or Q^H or P or P^H is applied to C from the left;
if SIDE = 'R', then Q or Q^H or P or P^H is applied to C from the right.

Constraint: SIDE = 'L' or 'R'.

3: TRANS — CHARACTER*1 *Input*

On entry: indicates whether Q or P or Q^H or P^H is to be applied to C as follows:

if TRANS = 'N', then Q or P is applied to C;
if TRANS = 'C', then Q^H or P^H is applied to C.

Constraint: TRANS = 'N' or 'C'.

4: M — INTEGER *Input*

On entry: m_C, the number of rows of the matrix C.

Constraint: M $\geq$ 0.

5: N — INTEGER *Input*

On entry: n_C, the number of columns of the matrix C.

Constraint: N $\geq$ 0.

6: K — INTEGER *Input*

On entry: if VECT = 'Q', the number of columns in the original matrix A; if VECT = 'P', the number of rows in the original matrix A.

Constraint: K $\geq$ 0.

7: A(LDA,$*$) — ***complex*** array *Input*

Note: the second dimension of the array A must be at least max(1,min(r,K})) if VECT = 'Q' and at least max$(1, r)$ if VECT = 'P'.

On entry: details of the vectors which define the elementary reflectors, as returned by F08KSF (CGEBRD/ZGEBRD).

8: LDA — INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F08KUF (CUNMBR/ZUNMBR) is called.

Constraints:

LDA $\geq$ max(1,r) if VECT = 'Q',
LDA $\geq$ max(1,min(r,K)) if VECT = 'P'.

9: TAU($*$) — ***complex*** array *Input*

Note: the dimension of the array TAU must be at least max(1,min(r,K)).

On entry: further details of the elementary reflectors, as returned by F08KSF (CGEBRD/ZGEBRD) in its parameter TAUQ if VECT = 'Q', or in its parameter TAUP if VECT = 'P'.

10: C(LDC,$*$) — ***complex*** array *Input*

Note: the second dimension of the array C must be at least max(1,N).

On entry: the matrix C.

On exit: C is overwritten by QC or Q^HC or CQ^H or CQ or PC or P^HC or CP^H or CP as specified by VECT, SIDE and TRANS.

11: LDC — INTEGER *Input*

On entry: the first dimension of the array C as declared in the (sub)program from which F08KUF (CUNMBR/ZUNMBR) is called.

Constraint: LDC $\geq$ max(1,M).

12: WORK(LWORK) — ***complex*** array *Input*

On exit: if INFO = 0, WORK(1) contains the minimum value of LWORK required for optimum performance.

13: LWORK — INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F08KUF (CUNMBR/ZUNMBR) is called.

Suggested value: for optimum performance LWORK should be at least N × nb if SIDE = 'L' and at least M × nb if SIDE = 'R', where nb is the ***blocksize***.

Constraints:

LWORK ≥ max(1,N) if SIDE = 'L',
LWORK ≥ max(1,M) if SIDE = 'R'.

14: INFO — INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed result differs from the exact result by a matrix E such that

$$\| E \|_2 = O(\epsilon)\| C \|_2,$$

where ϵ is the ***machine precision***.

8 Further Comments

The total number of real floating-point operations is approximately

$$\begin{array}{ll} 8n_C k(2m_C - k) & \text{if SIDE = 'L' and } m_C \geq k; \\ 8m_C k(2n_C - k) & \text{if SIDE = 'R' and } n_C \geq k; \\ 8m_C^2 n_C & \text{if SIDE = 'L' and } m_C < k; \\ 8m_C n_C^2 & \text{if SIDE = 'R' and } n_C < k; \end{array}$$

here k is the value of the parameter K.

The real analogue of this routine is F08KGF (SORMBR/DORMBR).

9 Example

For this routine two examples are presented. Both illustrate how the reduction to bidiagonal form of a matrix A may be preceded by a QR or LQ factorization of A.

In the first example, $m > n$, and

$$A = \begin{pmatrix} 0.96 - 0.81i & -0.03 + 0.96i & -0.91 + 2.06i & -0.05 + 0.41i \\ -0.98 + 1.98i & -1.20 + 0.19i & -0.66 + 0.42i & -0.81 + 0.56i \\ 0.62 - 0.46i & 1.01 + 0.02i & 0.63 - 0.17i & -1.11 + 0.60i \\ -0.37 + 0.38i & 0.19 - 0.54i & -0.98 - 0.36i & 0.22 - 0.20i \\ 0.83 + 0.51i & 0.20 + 0.01i & -0.17 - 0.46i & 1.47 + 1.59i \\ 1.08 - 0.28i & 0.20 - 0.12i & -0.07 + 1.23i & 0.26 + 0.26i \end{pmatrix}.$$

The routine first performs a QR factorization of A as $A = Q_a R$ and then reduces the factor R to bidiagonal form B: $R = Q_b B P^H$. Finally it forms Q_a and calls F08KUF (CUNMBR/ZUNMBR) to form $Q = Q_a Q_b$.

In the second example, $m < n$, and

$$A = \begin{pmatrix} 0.28 - 0.36i & 0.50 - 0.86i & -0.77 - 0.48i & 1.58 + 0.66i \\ -0.50 - 1.10i & -1.21 + 0.76i & -0.32 - 0.24i & -0.27 - 1.15i \\ 0.36 - 0.51i & -0.07 + 1.33i & -0.75 + 0.47i & -0.08 + 1.01i \end{pmatrix}.$$

The routine first performs an LQ factorization of A as $A = LP_a^H$ and then reduces the factor L to bidiagonal form B: $L = QBP_b^H$. Finally it forms P_b^H and calls F08KUF (CUNMBR/ZUNMBR) to form $P^H = P_b^H P_a^H$.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F08KUF Example Program Text
*     Mark 16 Release. NAG Copyright 1992.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          MMAX, NMAX, LDA, LDPH, LDU, LWORK
      PARAMETER        (MMAX=8,NMAX=8,LDA=MMAX,LDPH=NMAX,LDU=MMAX,
     +                 LWORK=64*(MMAX+NMAX))
      complex          ZERO
      PARAMETER        (ZERO=(0.0e0,0.0e0))
*     .. Local Scalars ..
      INTEGER          I, IC, IFAIL, INFO, J, M, N
*     .. Local Arrays ..
      complex          A(LDA,NMAX), PH(LDPH,NMAX), TAU(NMAX),
     +                 TAUP(NMAX), TAUQ(NMAX), U(LDU,NMAX), WORK(LWORK)
      real             D(NMAX), E(NMAX-1)
      CHARACTER        CLABS(1), RLABS(1)
*     .. External Subroutines ..
      EXTERNAL         F06TFF, F06THF, X04DBF, cgebrd, cgelqf, cgeqrf,
     +                 cunglq, cungqr, cunmbr
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F08KUF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      DO 20 IC = 1, 2
         READ (NIN,*) M, N
         IF (M.LE.MMAX .AND. N.LE.NMAX) THEN
*
*           Read A from data file
*
            READ (NIN,*) ((A(I,J),J=1,N),I=1,M)
*
            IF (M.GE.N) THEN
*
*              Compute the QR factorization of A
*
               CALL cgeqrf(M,N,A,LDA,TAU,WORK,LWORK,INFO)
*
*              Copy A to U
*
```

```
               CALL F06TFF('Lower',M,N,A,LDA,U,LDU)
*
*              Form Q explicitly, storing the result in U
*
               CALL cungqr(M,M,N,U,LDU,TAU,WORK,LWORK,INFO)
*
*              Copy R to PH (used as workspace)
*
               CALL F06TFF('Upper',N,N,A,LDA,PH,LDPH)
*
*              Set the strictly lower triangular part of R to zero
*
               CALL F06THF('Lower',N-1,N-1,ZERO,ZERO,PH(2,1),LDPH)
*
*              Bidiagonalize R
*
               CALL cgebrd(N,N,PH,LDPH,D,E,TAUQ,TAUP,WORK,LWORK,INFO)
*
*              Update Q, storing the result in U
*
               CALL cunmbr('Q','Right','No transpose',M,N,N,PH,LDPH,
     +                     TAUQ,U,LDU,WORK,LWORK,INFO)
*
*              Print bidiagonal form and matrix Q
*
               WRITE (NOUT,*)
               WRITE (NOUT,*) 'Example 1: bidiagonal matrix B'
               WRITE (NOUT,*) 'Diagonal'
               WRITE (NOUT,99999) (D(I),I=1,N)
               WRITE (NOUT,*) 'Super-diagonal'
               WRITE (NOUT,99999) (E(I),I=1,N-1)
               WRITE (NOUT,*)
               IFAIL = 0
*
               CALL X04DBF('General',' ',M,N,U,LDU,'Bracketed','F7.4',
     +                     'Example 1: matrix Q','Integer',RLABS,
     +                     'Integer',CLABS,80,0,IFAIL)
*
            ELSE
*
*              Compute the LQ factorization of A
*
               CALL cgelqf(M,N,A,LDA,TAU,WORK,LWORK,INFO)
*
*              Copy A to PH
*
               CALL F06TFF('Upper',M,N,A,LDA,PH,LDPH)
*
*              Form Q explicitly, storing the result in PH
*
               CALL cunglq(N,N,M,PH,LDPH,TAU,WORK,LWORK,INFO)
*
*              Copy L to U (used as workspace)
*
               CALL F06TFF('Lower',M,M,A,LDA,U,LDU)
*
*              Set the strictly upper triangular part of L to zero
*
```

```
               CALL F06THF('Upper',M-1,M-1,ZERO,ZERO,U(1,2),LDU)
*
*              Bidiagonalize L
*
               CALL cgebrd(M,M,U,LDU,D,E,TAUQ,TAUP,WORK,LWORK,INFO)
*
*              Update P**H, storing the result in PH
*
               CALL cunmbr('P','Left','Conjugate transpose',M,N,M,U,LDU,
     +                     TAUP,PH,LDPH,WORK,LWORK,INFO)
*
*              Print bidiagonal form and matrix P**H
*
               WRITE (NOUT,*)
               WRITE (NOUT,*) 'Example 2: bidiagonal matrix B'
               WRITE (NOUT,*) 'Diagonal'
               WRITE (NOUT,99999) (D(I),I=1,M)
               WRITE (NOUT,*) 'Super-diagonal'
               WRITE (NOUT,99999) (E(I),I=1,M-1)
               WRITE (NOUT,*)
               IFAIL = 0
*
               CALL X04DBF('General',' ',M,N,PH,LDPH,'Bracketed','F7.4',
     +                     'Example 2: matrix P**H','Integer',RLABS,
     +                     'Integer',CLABS,80,0,IFAIL)
*
            END IF
         END IF
   20 CONTINUE
      STOP
*
99999 FORMAT (3X,(8F8.4))
      END
```

9.2 Program Data

```
F08KUF Example Program Data
  6  4                                        :Values of M and N, Example 1
 ( 0.96,-0.81) (-0.03, 0.96) (-0.91, 2.06) (-0.05, 0.41)
 (-0.98, 1.98) (-1.20, 0.19) (-0.66, 0.42) (-0.81, 0.56)
 ( 0.62,-0.46) ( 1.01, 0.02) ( 0.63,-0.17) (-1.11, 0.60)
 (-0.37, 0.38) ( 0.19,-0.54) (-0.98,-0.36) ( 0.22,-0.20)
 ( 0.83, 0.51) ( 0.20, 0.01) (-0.17,-0.46) ( 1.47, 1.59)
 ( 1.08,-0.28) ( 0.20,-0.12) (-0.07, 1.23) ( 0.26, 0.26)   :End of matrix A
  3  4                                        :Values of M and N, Example 2
 ( 0.28,-0.36) ( 0.50,-0.86) (-0.77,-0.48) ( 1.58, 0.66)
 (-0.50,-1.10) (-1.21, 0.76) (-0.32,-0.24) (-0.27,-1.15)
 ( 0.36,-0.51) (-0.07, 1.33) (-0.75, 0.47) (-0.08, 1.01)   :End of matrix A
```

9.3 Program Results

```
F08KUF Example Program Results

Example 1: bidiagonal matrix B
Diagonal
   -3.0870 -2.0660 -1.8731 -2.0022
Super-diagonal
    2.1126 -1.2628  1.6126

Example 1: matrix Q
                   1                 2                 3                 4
1  (-0.3110, 0.2624) ( 0.6521, 0.5532) ( 0.0427, 0.0361) (-0.2634,-0.0741)
2  ( 0.3175,-0.6414) ( 0.3488, 0.0721) ( 0.2287, 0.0069) ( 0.1101,-0.0326)
3  (-0.2008, 0.1490) (-0.3103, 0.0230) ( 0.1855,-0.1817) (-0.2956, 0.5648)
4  ( 0.1199,-0.1231) (-0.0046,-0.0005) (-0.3305, 0.4821) (-0.0675, 0.3464)
5  (-0.2689,-0.1652) ( 0.1794,-0.0586) (-0.5235,-0.2580) ( 0.3927, 0.1450)
6  (-0.3499, 0.0907) ( 0.0829,-0.0506) ( 0.3202, 0.3038) ( 0.3174, 0.3241)

Example 2: bidiagonal matrix B
Diagonal
    2.7615  1.6298 -1.3275
Super-diagonal
   -0.9500 -1.0183

Example 2: matrix P**H
                   1                 2                 3                 4
1  (-0.1258, 0.1618) (-0.2247, 0.3864) ( 0.3460, 0.2157) (-0.7099,-0.2966)
2  ( 0.4148, 0.1795) ( 0.1368,-0.3976) ( 0.6885, 0.3386) ( 0.1667,-0.0494)
3  ( 0.4575,-0.4807) (-0.2733, 0.4981) (-0.0230, 0.3861) ( 0.1730, 0.2395)
```